Mathematics Notes

Note 25

August 1972

Analytical Approximations and Numerical Techniques for the Integral of the Anger-Weber Function

Carl E. Baum
Air Force Weapons Laboratory

H. Chang
Joe P. Martinez
The Dikewood Corporation
Albuquerque, New Mexico 87106

Abstract

A series expression for small arguments and an asymptotic expression for large arguments have been derived for the integral of the Anger-Weber function. Tables and graphs are given for various values of order and argument. Numerical methods employed to reduce computer time and error are discussed.

I. INTRODUCTION

The integral of the Anger-Weber function, in the following special form,

$$S_m(z) = \frac{i}{2}\int_0^{2z}\left[\mathbf{J}_m(\zeta) + i\,\mathbf{E}_m(\zeta)\right]d\zeta \tag{1}$$

is commonly encountered in the study of toroidal structures. Collin and Zucker[1] discuss this function and list several references which relate to it. An example of a structure of this nature is the TORUS type of EMP simulator.

Anger's function is defined to be[2]

$$\mathbf{J}_\nu(z) = \frac{1}{\pi}\int_0^{\pi}\cos(\nu\phi - z\sin(\phi))\,d\phi \tag{2}$$

This reduces to the Bessel function $J_\nu(z)$ if ν is an integer.

Weber's function is

$$\mathbf{E}_\nu(z) = \frac{1}{\pi}\int_0^{\pi}\sin(\nu\phi - z\sin(\phi))\,d\phi \tag{3}$$

This is also written as $-\Omega(z)$ where $\Omega(z)$ is the Lommel-Weber function referred to by Wu.[3]

---

[1] Collin, R. E., and F. J. Zucker, Antenna Theory, Part 1, McGraw-Hill, New York, 1969, Chapter 11.

[2] Abramowitz, M., and I. A. Stegun, editors, Handbook of Mathematical Functions, AMS 55, National Bureau of Standards, 1964, p. 498.

[3] Wu, T. T., "Theory of the Thin Circular Loop Antenna," J. Math. Phys., Vol. 3, 1962, pp. 1301-1304.

In this study ν is an integer, and designated as m. The Anger and Weber functions may be combined and written as

$$\mathbf{J}_m(\zeta) + i\mathbf{E}_m(\zeta) = \frac{1}{\pi}\int_0^{\pi} e^{i[m\phi - \zeta\sin(\phi)]}d\phi$$

$$= \frac{1}{\pi}\int_0^{\pi/2}[e^{im\phi} + e^{im\pi}e^{-im\phi}]e^{-i\zeta\sin(\phi)}d\phi \qquad (4)$$

Substituting this in Eq. 1

$$S_m(z) = \frac{i}{2\pi}\int_0^{2z}\int_0^{\pi/2}[e^{im\phi} + e^{im\pi}e^{-im\phi}]e^{-i\zeta\sin(\phi)}d\phi\, d\zeta \qquad (5)$$

Interchanging the order of integration and performing the second integration one gets

$$S_m(z) = \frac{1}{2\pi}\int_0^{\pi/2}\frac{[e^{im\phi} + e^{im\pi}e^{-im\phi}]}{\sin(\phi)}\left[1 - e^{-i2z\sin(\phi)}\right]d\phi \qquad (6)$$

which is the form this note will work on.

For numerical integration, the real and imaginary parts may be separated. For even m,

$$S_m(z) = \frac{1}{\pi}\int_0^{\pi/2}\frac{\cos(m\phi)}{\sin(\phi)}\Big(1 - \cos(2z\sin(\phi))\Big)\,d\phi + i\frac{1}{\pi}\int_0^{\pi/2}\frac{\cos(m\phi)}{\sin(\phi)}\sin(2z\sin(\phi))\,d\phi \qquad (7)$$

and for odd m,

$$S_m(z) = -\frac{1}{\pi}\int_0^{\pi/2} \frac{\sin(m\phi)}{\sin(\phi)} \sin(2z\sin(\phi))\,d\phi + i\frac{1}{\pi}\int_0^{\pi/2} \frac{\sin(m\phi)}{\sin(\phi)}\Big(1 - \cos(2z\sin(\phi))\Big)d\phi \qquad (8)$$

In all calculations in this note only real z is considered. On the computer, certain checks may be made to determine if m is even or odd in order to evaluate the correct integral. The real and imaginary parts are then combined into a complex variable to give the value for $S_m(z)$. The integration method used in this note is the Gaussian Quadrature method of order 40.[4] The subroutine is set to calculate the integral such that the relative difference between two successive calculations is in the order of 10^{-5}. This method proves to be very accurate, but it is slow on the computer.

[4] Ref. 2, pp. 887, 888, 917.

II. SERIES EXPRESSION FOR SMALL ARGUMENTS

For small z the exponentials may be expanded into a converging series, so that

$$S_m(z) = -\frac{1}{2\pi}\int_0^{\pi} e^{im\phi} \sum_{\ell=1}^{\infty} (-i2z)^{\ell} \frac{(\sin(\phi))^{\ell-1}}{\ell!} d\phi$$

$$= \sum_{\ell=0}^{\infty} \frac{-1}{2\pi} \frac{(-i2z)^{\ell+1}}{(\ell+1)!} q_{m,\ell} \tag{9}$$

where

$$q_{m,\ell} = \int_0^{\pi} e^{im\phi} (\sin(\phi))^{\ell} d\phi$$

$$= \frac{\pi}{2^{\ell}} \frac{\Gamma(\ell+1)\, e^{i\frac{\pi m}{2}}}{\Gamma\left(\frac{\ell+m}{2}+1\right)\Gamma\left(\frac{\ell-m}{2}+1\right)} \tag{10}$$

is derived in Appendix A.

Since m and ℓ are integers, the equations above may be further reduced to

$$S_m(z) = \sum_{\ell=0}^{\infty} P_{m,\ell}\, z^{\ell+1} \tag{11}$$

where

$$P_{m,\ell} = \frac{i^{m-\ell+1}}{(\ell+1)\,\Gamma\left(\frac{\ell+m}{2}+1\right)\Gamma\left(\frac{\ell-m}{2}+1\right)} \tag{12}$$

In doing the computer calculations, the Gamma functions were calculated only once and then stored in two arrays. One array containing the function for whole number arguments and the other for arguments of odd multiples of 1/2. Using the initial values

$$\Gamma(\tfrac{1}{2}) = \sqrt{\pi}$$

$$\Gamma(1) = 1 \qquad (13)$$

$$\Gamma(2) = 1$$

and the recurrence formula,

$$\Gamma(x + 1) = x\Gamma(x) \qquad (14)$$

the functions were calculated for positive integers to $\Gamma(176)$ and for half-value arguments from $\Gamma(-\frac{21}{2})$ to $\Gamma(+\frac{351}{2})$.

It was found that in doing the computer calculations of $S_m(z)$ from Eq. 11 directly, round-off error was significant for values of z larger than 15. This is because $z^{\ell+1}$ gets large faster than the coefficient at first, causing very large numbers which succeeding terms cannot subtract because bits are truncated. The machine used is the Control Data Corporation 6600, which has an accuracy of about 14 digits with the allowable range of 10^{-295} to 10^{322}. The accuracy of $S_m(z)$ using the series expression as compared to the integration and a convergence criteria of

$$\left| P_{m,n} z^{n+1} \Big/ \sum_{\ell=0}^{n-1} P_{m,\ell} z^{\ell+1} \right| \le 10^{-7} \qquad (15)$$

was within .001% for z up to 15, but deteriorated seriously beyond this point. Also the complex arithmetic caused the calculations to be slower. The value of ℓ could not proceed beyond about 100 because in the machine complex division is accomplished by taking complex conjugates, thereby squaring each gamma function and exceeding machine limits sooner than necessary.

The series may be calculated without complex arithmetic as shown in the next equation. This modification makes the individual terms more like each other in terms of magnitude.

$$S_m(z) = i^m \left[\sum_{\ell=1}^{\infty,4} (b_\ell - b_{\ell+2} z^2) z^{\ell+1} + i \sum_{\ell=0}^{\infty,4} (b_\ell - b_{\ell+2} z^2) z^{\ell+1} \right] \tag{16}$$

where now

$$b_\ell = \frac{1}{(\ell + 1)\Gamma\left(\frac{\ell+m}{2} + 1\right)\Gamma\left(\frac{\ell-m}{2} + 1\right)} \tag{17}$$

This has the added result that one pass through a do-loop will calculate four terms, reducing the time somewhat. Since b_ℓ is not dependent on z one may calculate the b_ℓ coefficients only once for each m and store them in an array to be used with many values of z.

Using double precision and a convergence criteria on each sum as before the usable range was increased to z = 33. The accuracy was within .001% as compared to the direct integration method. The time was from 4 to 40 times faster, a very significant savings.

III. ASYMPTOTIC EXPANSION FOR LARGE ARGUMENTS

From Eq. 6 it can be seen that the function may be written as

$$S_m(z) = S_{m_1}(z) + S_{m_2}(z) \tag{18}$$

where

$$S_{m_1}(z) = \frac{1}{2\pi}\int_0^{\pi/2} \frac{\left[e^{im\phi} + e^{im\pi}e^{-im\phi}\right]\left[1 - e^{-i2z\phi}\right] d\phi}{\sin(\phi)} \tag{19}$$

and

$$S_{m_2}(z) = \frac{1}{2\pi}\int_0^{\pi/2} \frac{\left[e^{im\phi} + e^{im\pi}e^{-im\phi}\right]\left[e^{-i2z\phi} - e^{-i2z\sin(\phi)}\right] d\phi}{\sin(\phi)} \tag{20}$$

Note that $\frac{1 - e^{-i2z\phi}}{\sin(\phi)}$ and $\frac{e^{-i2z\phi} - e^{-i2z\sin(\phi)}}{\sin(\phi)}$ are bounded, even when $\phi \to 0$.

In Appendix B certain relationships are derived which will contribute to the development of the asymptotic expansion. From Eqs. (B-14) and (B-15) in that appendix it can be shown that,

$$S_{m_1}(z) = \frac{1}{2\pi}\int_0^{\pi/2}\left(\frac{e^{im\phi} - e^{-i(2z-m)\phi}}{\sin(\phi)}\right)d\phi + \frac{e^{im\pi}}{2\pi}\int_0^{\pi/2}\left(\frac{e^{-im\phi} - e^{-i(2z+m)\phi}}{\sin(\phi)}\right)d\phi$$

$$= \frac{1}{4\pi}\left(e^{i\pi\left(\frac{m-1}{2}\right)} G\left(\frac{m+1}{2}\right) - e^{-i\pi\left(z-\frac{m+1}{2}\right)} G\left(z - \frac{m-1}{2}\right)\right) + \frac{1}{2\pi}\left(\psi\left(z - \frac{m-1}{2}\right)\right.$$

$$\left. - \psi\left(\frac{m+1}{2}\right)\right) + \frac{i}{2} + e^{im\pi}\left\{\frac{1}{4\pi}\left(e^{-i\pi\left(\frac{m-1}{2}\right)} G\left(\frac{m+1}{2}\right) - e^{-i\pi\left(z+\frac{m-1}{2}\right)}\right.\right.$$

$$\left.\left. G\left(z + \frac{m+1}{2}\right)\right) + \frac{1}{2\pi}\left(\psi\left(z + \frac{m+1}{2}\right) - \psi\left(\frac{m+1}{2}\right)\right)\right\}$$

$$= \frac{i}{2} - \frac{1}{4\pi} e^{-i\pi\left(z - \frac{m+1}{2}\right)} \left(G\left(z - \frac{m-1}{2}\right) + G\left(z + \frac{m+1}{2}\right) \right) + \frac{1}{2\pi} \left(\psi\left(z - \frac{m-1}{2}\right) \right.$$

$$\left. + e^{im\pi} \psi\left(z + \frac{m+1}{2}\right) \right) - \frac{1}{2\pi}\left(1 + e^{im\pi}\right) \psi\left(\frac{m+1}{2}\right) \tag{21}$$

Considering the behavior for large z, it has been shown[5] that

$$\psi(z) = \ln(z) - \frac{1}{2z} + O(z^{-2}) \tag{22}$$

From Eq. (B-13) in Appendix B,

$$G(\alpha) = \psi\left(\frac{\alpha+1}{2}\right) - \psi\left(\frac{\alpha}{2}\right)$$

$$= \ln\left(\frac{\alpha+1}{2}\right) - \frac{1}{\alpha+1} - \ln\left(\frac{\alpha}{2}\right) + \frac{1}{\alpha} + O(\alpha^{-2}) \tag{23}$$

$$= \ln\left(1 + \frac{1}{\alpha}\right) + \frac{1}{\alpha(\alpha+1)} + O(\alpha^{-2}) = \frac{1}{\alpha} + O(\alpha^{-2}) = O(\alpha^{-1})$$

Therefore,

$$S_{m_1}(z) = \frac{i}{2} + \frac{1}{2\pi}\left(\ln\left(z - \frac{m-1}{2}\right) + e^{im\pi} \ln\left(z + \frac{m+1}{2}\right) \right)$$

$$- \frac{1}{2\pi}\left(1 + e^{im\pi}\right) \psi\left(\frac{m+1}{2}\right) + O\left(\frac{1}{z}\right) \tag{24}$$

This may be further simplified if the log terms are considered such that

$$\ln\left(z - \frac{m-1}{2}\right) + e^{im\pi} \ln\left(z + \frac{m+1}{2}\right) = \ln\left[z\left(1 - \frac{m-1}{2z}\right)\right] + (-1)^m \ln\left[z\left(1 + \frac{m+1}{2z}\right)\right] \tag{25}$$

[5] Ref. (2), p. 259, Eq. 6.3.18.

For small α,

$$\ln(1+\alpha) \simeq \alpha \tag{26}$$

so, for m even

$$\ln\left[z\left(1-\frac{m-1}{2z}\right)\right] + \ln\left[z\left(1+\frac{m+1}{2z}\right)\right] \simeq 2\ln z - \frac{m-1}{2z} + \frac{m+1}{2z} = 2\ln z + O\left(\frac{1}{z}\right) \tag{27}$$

and for m odd

$$\begin{aligned}\ln\left[z\left(1-\frac{m-1}{2z}\right)\right] - \ln\left[z\left(1+\frac{m+1}{2z}\right)\right] &\simeq -\frac{m-1}{2z} - \frac{m+1}{2z} \\ &= -\frac{m}{z} + O\left(\frac{1}{z}\right) \\ &= O\left(\frac{1}{z}\right)\end{aligned} \tag{28}$$

Eq. 24 now becomes

$$\begin{aligned}S_{m_1}(z) &= \frac{i}{2} + \frac{1}{2\pi}\left[1+(-1)^m\right]\ln z - \frac{1}{2\pi}\left[1+(-1)^m\right]\psi\left(\frac{m+1}{2}\right) + O\left(\frac{1}{z}\right) \\ &= \frac{i}{2} + \frac{1}{2\pi}\left[(1+(-1)^m)\left(\ln z - \psi\left(\frac{m+1}{2}\right)\right)\right] + O\left(\frac{1}{z}\right)\end{aligned} \tag{29}$$

Note that when m is odd only $\frac{i}{2}$ contributes. When m is even a real part is also present.

Now consider S_{m_2} of Eq. 20. Since,

$$\lim_{\phi \to 0} \frac{e^{-i2z\phi} - e^{-i2z\sin(\phi)}}{\sin(\phi)} = 0 \tag{30}$$

it may be said that

$$S_{m_2}(z) = \frac{1}{2\pi}\int_{0+}^{\pi/2} \left(\frac{e^{im\phi} + e^{im\pi}e^{-im\phi}}{\sin(\phi)}\right)\left(e^{-i2z\phi} - e^{-i2z\sin(\phi)}\right) \qquad (31)$$

That is, the integration with respect to ϕ does not include the point $\phi = 0$. The term

$$\frac{e^{im\phi} + e^{im\pi}e^{-im\phi}}{\sin(\phi)}$$

is an analytic function in the region $0 < \phi \le \frac{\pi}{2}$.

From Copson,[6] if f(x) has no stationary point in $\alpha \le x \le \beta$, then

$$I = \int_{\alpha}^{\beta} e^{i\nu f(x)}\,\phi(x)\,dx$$

$$= \frac{\phi(\beta)}{i\nu f'(\beta)}\, e^{i\nu f(\beta)} - \frac{\phi(\alpha)}{i\nu f'(\alpha)}\, e^{i\nu f(\alpha)} + O\left(\frac{1}{\nu^2}\right) \qquad (32)$$

$$= O\left(\frac{1}{\nu}\right) \text{as } \nu \to \infty$$

Therefore,

$$\int_{0+}^{\pi/2} \frac{e^{im\phi} + e^{im\pi}e^{-im\phi}}{\sin(\phi)}\, e^{-i2z\phi}\, d\phi = O\left(\frac{1}{z}\right) \qquad (33)$$

since 2ϕ has no stationary point in $0 < \phi \le \frac{\pi}{2}$.

---

[6] Copson, E. T., Asymptotic Expansions, Cambridge University Press, 1965, Chapter 4.

If, on the other hand, f(x) has one stationary point in $\alpha \leq x \leq \beta$, namely at $x = \beta$, and if $f''(\beta) > 0$, then

$$I = \int_{\alpha}^{\beta} e^{i\nu f(x)} \phi(x)\, dx = \left(\frac{\pi}{2\nu f''(\beta)}\right)^{1/2} \phi(\beta) e^{i(\nu f(\beta) + \frac{1}{4}\pi)} + O\left(\frac{1}{\nu}\right) \qquad (34)$$

Thus,

$$S_{m_2}(z) = O\left(\frac{1}{z}\right) - \frac{1}{2\pi}\int_{0+}^{\pi/2} \frac{e^{im\phi} + e^{im\pi} e^{-im\phi}}{\sin(\phi)} e^{-i2z\sin(\phi)} d\phi$$

$$= -\frac{1}{2}(\pi z)^{-1/2} e^{i\left(\frac{m\pi}{2} - 2z + \frac{\pi}{4}\right)} + O\left(\frac{1}{z}\right) \qquad (35)$$

because $f(\phi) = -2\sin(\phi)$ has a stationary point at $\phi = \frac{\pi}{2}$ where $f''\left(\frac{\pi}{2}\right) > 0$.

Now $S_{m_1}(z)$ and $S_{m_2}(z)$ for large z may be combined to give the asymptotic form.

In doing the numerical calculations for the asymptotic form, $S_{m_2}(z)$ of Eq. 35 was combined with $S_{m_1}(z)$ as given in Eq. 29 as well as in Eq. 24. It was found that Eq. 24 was more accurate for wider ranges of z and m, particularly when m is odd and large. It can be seen from Eq. 28 that $\frac{m}{z}$ is not in the order of 1/z if the size of m is comparable to that of z. So for the purposes of the computer program the asymptotic form used is, for large z,

$$S_m(z) = S_{m_1}(z) + S_{m_2}(z) = \frac{i}{2} + \frac{1}{2\pi}\left[\ln\left(z - \frac{m-1}{2}\right) + (-1)^m \ln\left(z + \frac{m+1}{2}\right)\right]$$

$$- \frac{1}{2\pi}\left[1 + (-1)^m\right]\psi\left(\frac{m+1}{2}\right)$$

$$- \frac{1}{2}(\pi z)^{-1/2} e^{i(2m+1)\frac{\pi}{4}} e^{-i2z} + O\left(\frac{1}{z}\right) \qquad (36)$$

Although not used in this note for numerical work, it may be reduced to

$$S_m(z) = \frac{i}{2} + \frac{1}{2\pi}\left(1 + (-1)^m\right)\left(\ln z - \psi\left(\frac{m+1}{2}\right)\right) - \frac{1}{2}(\pi z)^{-1/2} e^{i\left[(2m+1)\frac{\pi}{4} - 2z\right]} + O\left(\frac{1}{z}\right) \qquad (37)$$

In the numerical calculation of $\psi\left(\frac{m+1}{2}\right)$, the initial value and recurrence formula[7]

$$\psi\left(\frac{1}{2}\right) = -\gamma - 2\ln 2 = -1.963510026021423$$

$$\psi(z+1) = \psi(z) + \frac{1}{z} \tag{38}$$

were used.

The error in using Eq. 36 will be on the order of $1/z$. If z is large enough this asymptotic formula will be sufficient to represent $S_m(z)$ accurately. Calculations of Eq. 36 show that the computer computation time is from 100 to 1000 times faster than when using the direct integration method. The tables in Section IV show $S_m(z)$ as calculated by the direct integration method, series method for small argument and asymptotic form for large argument, as well as relative differences. Computer central processor elapsed time is also shown.

[7] Ref. (2), p. 258, Eqs. 6.3.3 and 6.3.5.

IV. TABLES, PLOTS, AND SUBROUTINE USE

At the end of this section there are tables of $S_m(z)$ calculated by the direct integration method and by an approximating form (series or asymptotic) for $0.1 \leq z \leq 1000.0$ and for m = 0, 1, 10, 19, and 20. The computer elapsed time for each calculation is given. In many cases the figure given is zero. This is because the clock in the computer is accurate only to 1/1000 of a second; so, it may be assumed that the elapsed time for these calculations are <.001 sec. The column labeled "%" is the percent difference between the integration method calculations, $S_{m_{int}}(z)$, and the approximation calculations, $S_{m_{approx.}}(z)$. This is determined by the formula

$$\% \text{ difference} = \left| \frac{S_{m_{int}}(z) - S_{m_{approx}}(z)}{S_{m_{int}}(z)} \right| \times 100 \qquad (39)$$

It can be seen from the tables that the series approximation is very good to z = 30. But as was discussed in Section II, round-off error becomes significant at z = 33, rendering the series method useless for values of z larger than this. So in the tables the approximation form z = 33 to z = 1000 is that of the asymptotic expansion.

The graphs at the end of this section are for m = 3, 4, 17, and 18. The values picked for m for the tables and graphs were some which would depict a fairly wide range, so that the behavior of $S_m(z)$ may be seen through this range.

The solid lines on the graphs indicate the value of $S_m(z)$ calculated by the direct integration method, and the dashed lines indicate the series or asymptotic values. The series method is used for z < 33 and the asymptotic form for $z \geq 33$. The difference is so slight for small m that the curves overlap. For large m there is only a small range where the difference is significant. The subroutine which calculates $S_m(z)$ does so by the series method to z = 33 and then switches to the integration method. It carries on using the integration method until the point where the asymptotic form is within 1% of the integration calculations. For large m the integration method must go on to larger values of z where this occurs. The values of z for the different m where this happens are stored in a DATA statement in the subroutine ACCUR for m up to and including 20, so that the switch-over occurs at the proper place and no extra computer time is spent in searching. Subroutine ACCUR will search for values of z for any accuracy desired and for larger m also if the values of m exceed 20 and accuracies other than 1% are desired. This is done by the Fortran call

```
CALL ACCUR (N,ERR)
```

where N is the largest m desired and ERR is the largest relative difference desired between the integration and asymptotic methods. The switch-over values will be stored in an array and passed to SM in a COMMON statement. If no call to ACCUR is made then the default is N = 20 and ERR = .01. The relative difference is computed from Eq. 39 without the percent factor 100.

Appendix C contains a listing of the subroutine COMPLEX FUNCTION SM(M,Z) and associated subroutines which calculate $S_m(z)$. The use of the subroutine is fairly easy. Since it is a Fortran function it is used in an arithmetic replacement statement such as

```
A = SM(M,Z)
```

or

```
A = X*SM(M,Z)/COS(Y)
```

where SM and A have been typed COMPLEX and M and Z have been previously defined.

z	$S_{m_{int}}(z)$ Real	$S_{m_{int}}(z)$ Imaginary	$S_{m_{approx}}(z)$ Real	$S_{m_{approx}}(z)$ Imaginary	Time int	Time approx	Error %
			By Series				
.10	.00635	.09967	.00635	.09967	.040	.000	.000
.20	.02524	.19735	.02524	.19735	.038	.002	.000
.30	.05616	.29112	.05616	.29112	.038	.000	.000
.40	.09830	.37917	.09830	.37917	.041	.001	.000
.50	.15055	.45987	.15055	.45987	.038	.001	.000
.60	.21154	.53178	.21154	.53178	.041	.001	.000
.70	.27970	.59375	.27970	.59375	.040	.000	.000
.80	.35329	.64491	.35329	.64491	.038	.001	.000
.90	.43048	.68470	.43048	.68470	.038	.001	.000
1.00	.50935	.71289	.50935	.71289	.038	.001	.000
2.00	1.04577	.51237	1.04577	.51237	.037	.002	.000
3.00	.91574	.35311	.91574	.35311	.037	.002	.000
4.00	.97727	.60537	.97727	.60537	.038	.001	.000
5.00	1.25948	.53351	1.25948	.53351	.038	.003	.000
6.00	1.17726	.38706	1.17726	.38706	.038	.003	.000
7.00	1.15781	.56034	1.15781	.56034	.039	.002	.000
8.00	1.37321	.55043	1.37321	.55043	.039	.003	.000
9.00	1.33412	.40665	1.33412	.40665	.039	.003	.000
10.00	1.27423	.52919	1.27423	.52919	.038	.004	.000
20.00	1.57421	.56289	1.57421	.56289	.038	.006	.000
30.00	1.75321	.52405	1.75328	.52405	.039	.008	.004
			Asymptotic Form				
40.00	1.83428	.47241	1.83799	.47214	.039	.000	.197
50.00	1.86046	.46133	1.86337	.46139	.065	.000	.152
60.00	1.89240	.49380	1.89500	.49399	.090	.000	.133
70.00	1.95857	.52800	1.96095	.52809	.090	.000	.117
80.00	2.03661	.52672	2.03869	.52665	.089	.001	.099
90.00	2.08679	.49584	2.08853	.49574	.192	.001	.081
100.00	2.09867	.47289	2.10017	.47286	.191	.001	.070
200.00	2.33093	.49544	2.33172	.49541	.397	.002	.033
300.00	2.45156	.51203	2.45210	.51201	.398	.000	.022
400.00	2.52769	.51338	2.52810	.51339	.818	.000	.016
500.00	2.59078	.50235	2.59110	.50236	.809	.000	.012
600.00	2.65382	.49117	2.65408	.49117	.812	.001	.010
700.00	2.71409	.49004	2.71431	.49004	1.217	.000	.008
800.00	2.76265	.49857	2.76285	.49857	1.642	.001	.007
900.00	2.79598	.50747	2.79616	.50747	1.629	.000	.006
1000.00	2.82026	.50818	2.82042	.50818	1.629	.000	.006

Table 1. $S_m(z)$ for $m = 0$.

z	$S_{m_{int}}(z)$ Real	$S_{m_{int}}(z)$ Imaginary	$S_{m_{approx}}(z)$ Real	$S_{m_{approx}}(z)$ Imaginary	Time int	Time approx	Error %
			By Series				
.10	-.06338	.00499	-.06338	.00499	.039	.001	.000
.20	-.12507	.01980	-.12507	.01980	.039	.001	.000
.30	-.18346	.04400	-.18346	.04400	.038	.001	.000
.40	-.23700	.07686	-.23700	.07686	.039	.001	.000
.50	-.28433	.11740	-.28433	.11740	.039	.001	.000
.60	-.32427	.16443	-.32427	.16443	.040	0.000	.000
.70	-.35590	.21657	-.35590	.21657	.040	.001	.000
.80	-.37851	.27230	-.37851	.27230	.040	.001	.000
.90	-.39173	.33001	-.39173	.33001	.040	.001	.000
1.00	-.39543	.38805	-.39543	.38805	.039	.001	.000
2.00	-.06751	.69857	-.06751	.69857	.040	.001	.000
3.00	.09228	.42468	.09228	.42468	.041	.002	.000
4.00	-.15099	.41417	-.15099	.41417	.039	.002	.000
5.00	-.05937	.62297	-.05937	.62297	.039	.003	.000
6.00	.08627	.47616	.08627	.47616	.041	.002	.000
7.00	-.08622	.41446	-.08622	.41446	.040	.003	.000
8.00	-.06772	.58745	-.06772	.58745	.043	.003	.000
9.00	.07615	.50668	.07615	.50668	.039	.003	.000
10.00	-.04720	.41649	-.04720	.41649	.043	.004	.000
20.00	-.07092	.49632	-.07092	.49632	.042	.006	.000
30.00	-.02898	.54574	-.02898	.54573	.041	.009	.001
			Asymptotic Form				
40.00	.02383	.53487	.02393	.53483	.040	0.000	.021
50.00	.03544	.49001	.03546	.48996	.090	.001	.011
60.00	.00340	.46409	.00338	.46408	.092	.001	.004
70.00	-.03034	.48132	-.03034	.48135	.093	.001	.005
80.00	-.02865	.51684	-.02863	.51686	.091	0.000	.006
90.00	.00247	.52943	.00250	.52943	.144	.001	.006
100.00	.02554	.50772	.02555	.50770	.196	.001	.004
200.00	.00379	.51941	.00380	.51941	.404	.001	.002
300.00	-.01255	.51099	-.01254	.51100	.405	.001	.001
400.00	-.01378	.49555	-.01378	.49555	.829	.001	.000
500.00	-.00268	.48761	-.00268	.48761	.821	0.000	.000
600.00	.00857	.49261	.00857	.49261	.827	.001	.000
700.00	.00973	.50381	.00973	.50381	1.241	.001	.000
800.00	.00123	.50987	.00123	.50987	1.658	.002	.000
900.00	-.00765	.50571	-.00765	.50571	1.673	.001	.000
1000.00	-.00834	.49645	-.00834	.49645	1.665	.001	.000

Table 2. $S_m(z)$ for $m = 1$.

z	$S_{m_{int}}(z)$ Real	$S_{m_{int}}(z)$ Imaginary	$S_{m_{approx}}(z)$ Real	$S_{m_{approx}}(z)$ Imaginary	Time int	Time approx	Error %
			By Series				
.10	-.00006	.00000	-.00006	.00000	.040	.001	.000
.20	-.00026	.00000	-.00026	.00000	.039	.002	.000
.30	-.00058	.00000	-.00058	.00000	.040	.002	.000
.40	-.00103	.00000	-.00103	.00000	.041	.001	.000
.50	-.00162	.00000	-.00162	.00000	.040	.001	.000
.60	-.00233	.00000	-.00233	.00000	.041	.001	.000
.70	-.00319	.00000	-.00319	.00000	.041	.001	.000
.80	-.00417	.00000	-.00417	.00000	.040	.002	.000
.90	-.00530	.00000	-.00530	.00000	.039	.002	.000
1.00	-.00658	.00000	-.00658	.00000	.040	.001	.000
2.00	-.02840	.00004	-.02840	.00004	.040	.001	.000
3.00	-.07514	.00219	-.07514	.00219	.040	.002	.000
4.00	-.16838	.02918	-.16838	.02918	.040	.002	.000
5.00	-.29530	.15715	-.29530	.15715	.040	.002	.000
6.00	-.31177	.42872	-.31177	.42872	.040	.002	.000
7.00	-.07759	.64851	-.07759	.64851	.040	.002	.000
8.00	.19586	.56838	.19586	.56838	.040	.003	.000
9.00	.18110	.38953	.18110	.38953	.041	.003	.000
10.00	.07058	.46918	.07058	.46918	.040	.003	.000
20.00	.36939	.47494	.36939	.47494	.039	.006	.000
30.00	.51594	.51795	.51593	.51795	.040	.009	.002
			Asymptotic Form				
40.00	.64650	.54339	.62806	.52786	.040	.001	2.854
50.00	.75818	.52945	.74405	.53861	.092	.001	1.820
60.00	.82485	.49110	.82791	.50601	.093	.002	1.586
70.00	.84654	.46707	.85963	.47191	.094	0.000	1.444
80.00	.85717	.47967	.86652	.47335	.095	.001	1.149
90.00	.89178	.51210	.89136	.50426	.202	0.000	.764
100.00	.95181	.52825	.94654	.52714	.199	0.000	.495
200.00	1.15487	.50695	1.15496	.50459	.409	.001	.187
300.00	1.29075	.48893	1.29223	.48799	.410	0.000	.127
400.00	1.39790	.48636	1.39913	.48661	.830	0.000	.085
500.00	1.47759	.49703	1.47804	.49764	.857	.001	.049
600.00	1.53112	.50852	1.53103	.50883	.855	.001	.020
700.00	1.56899	.51009	1.56886	.50996	1.727	.001	.011
800.00	1.60512	.50173	1.60527	.50143	1.725	.001	..020
900.00	1.64652	.49269	1.64690	.49253	1.738	.001	.024
1000.00	1.68932	.49173	1.68968	.49182	1.766	.001	.021

Table 3. $S_m(z)$ for m = 10.

z	$S_{m_{int}}(z)$ Real	$S_{m_{int}}(z)$ Imaginary	$S_{m_{approx}}(z)$ Real	$S_{m_{approx}}(z)$ Imaginary	Time int	Time approx	Error %
			By Series				
.10	-.00335	.00000	-.00335	0.00000	.039	.001	.000
.20	-.00670	.00000	-.00670	0.00000	.039	0.000	.000
.30	-.01006	.00000	-.01006	0.00000	.039	.001	.000
.40	-.01341	.00000	-.01341	0.00000	.040	0.000	.000
.50	-.01677	.00000	-.01677	0.00000	.040	.001	.000
.60	-.02013	.00000	-.02013	0.00000	.041	.001	.000
.70	-.02350	.00000	-.02350	0.00000	.039	.001	.000
.80	-.02687	.00000	-.02687	0.00000	.040	.001	.000
.90	-.03025	.00000	-.03025	0.00000	.040	0.000	.000
1.00	-.03363	.00000	-.03363	0.00000	.040	0.000	.000
2.00	-.06804	.00000	-.06804	0.00000	.040	.001	.000
3.00	-.10413	.00000	-.10413	.00000	.040	.001	.000
4.00	-.14308	.00000	-.14308	.00000	.040	.001	.000
5.00	-.18673	.00001	-.18673	.00001	.040	.002	.000
6.00	-.23852	.00028	-.23852	.00028	.040	.002	.000
7.00	-.30564	.00317	-.30564	.00317	.040	.002	.000
8.00	-.39984	.02126	-.39984	.02126	.040	.003	.000
9.00	-.52261	.09072	-.52261	.09072	.040	.003	.000
10.00	-.62426	.25775	-.62426	.25775	.040	.003	.000
20.00	-.17894	.57476	-.17894	.57476	.040	.006	.000
30.00	-.13567	.54608	-.13567	.54607	.041	.008	.000
			Asymptotic Form				
40.00	-.08672	.54556	-.10395	.46517	.092	.001	14.883
50.00	-.04202	.53624	-.09921	.51004	.092	.001	11.695
60.00	-.01473	.50864	-.05641	.53592	.092	.001	9.790
70.00	-.01732	.47797	-.01507	.51865	.092	.001	8.520
80.00	-.04175	.46835	-.01109	.48314	.197	.001	7.240
90.00	-.06102	.48759	-.03780	.47057	.199	0.000	5.858
100.00	-.05339	.51658	-.05732	.49230	.199	.001	4.737
200.00	-.02773	.48449	-.01969	.48059	.409	0.000	1.842
300.00	-.00185	.48594	.00195	.48900	.620	.001	1.004
400.00	.00648	.50135	.00583	.50445	.832	.001	.631
500.00	-.00151	.51177	-.00368	.51239	.832	.001	.442
600.00	-.01267	.50863	-.01387	.50739	.832	.001	.338
700.00	-.01469	.49750	-.01428	.49619	1.676	0.000	.275
800.00	-.00631	.49035	-.00521	.49013	1.675	.001	.229
900.00	.00350	.49357	.00411	.49429	1.676	.001	.191
1000.00	.00545	.50280	.00516	.50355	1.676	.001	.159

Table 4. $S_m(z)$ for m = 19.

z	$S_{m_{int}}(z)$ Real	$S_{m_{int}}(z)$ Imaginary	$S_{m_{approx}}(z)$ Real	$S_{m_{approx}}(z)$ Imaginary	Time int	Time approx	Error %
			By Series				
.10	-.00002	.00000	-.00002	.00000	.039	.001	.000
.20	-.00006	.00000	-.00006	.00000	.039	.001	.000
.30	-.00014	.00000	-.00014	.00000	.039	.002	.000
.40	-.00026	.00000	-.00026	.00000	.039	.002	.000
.50	-.00040	.00000	-.00040	.00000	.039	.002	.000
.60	-.00058	.00000	-.00058	.00000	.041	.001	.000
.70	-.00078	.00000	-.00078	.00000	.039	.002	.000
.80	-.00102	.00000	-.00102	.00000	.040	.001	.000
.90	-.00130	.00000	-.00130	.00000	.039	.001	.000
1.00	-.00160	.00000	-.00160	.00000	.040	.001	.000
2.00	-.00652	.00000	-.00652	.00000	.040	.001	.000
3.00	-.01507	.00000	-.01507	.00000	.040	.002	.000
4.00	-.02789	.00000	-.02789	.00000	.040	.002	.000
5.00	-.04616	.00000	-.04616	.00000	.040	.002	.000
6.00	-.07211	.00009	-.07211	.00009	.041	.002	.000
7.00	-.11056	.00118	-.11056	.00118	.040	.002	.000
8.00	-.17095	.00943	-.17095	.00943	.040	.003	.000
9.00	-.26261	.04767	-.26261	.04767	.041	.002	.000
10.00	-.36698	.16082	-.36698	.16082	.041	.003	.000
20.00	.10078	.52390	.10078	.52390	.041	.006	.000
30.00	.27666	.48518	.27666	.48518	.040	.009	.000
			Asymptotic Form				
40.00	.38413	.50173	.46991	.47214	.093	.001	14.361
50.00	.47379	.52597	.49893	.46139	.092	.001	9.789
60.00	.56306	.53710	.53252	.49399	.092	0.000	6.790
70.00	.64149	.52285	.59965	.52809	.092	.001	5.096
80.00	.69027	.49239	.67815	.52665	.198	.001	4.287
90.00	.70669	.47149	.72850	.49574	.199	.001	3.839
100.00	.71244	.47865	.74051	.47286	.199	0.000	3.339
200.00	.96792	.48666	.97325	.49541	.408	.001	.945
300.00	1.09665	.50778	1.09385	.51201	.409	.001	.420
400.00	1.17297	.51408	1.16993	.51339	.830	.001	.243
500.00	1.23336	.50477	1.23296	.50236	.829	.001	.183
600.00	1.29434	.49251	1.29597	.49117	.831	0.000	.152
700.00	1.35453	.48960	1.35621	.49004	1.673	.001	.121
800.00	1.40430	.49735	1.40475	.49857	1.673	.001	.087
900.00	1.43868	.50679	1.43807	.50747	1.673	0.000	.060
1000.00	1.46301	.50850	1.46233	.50818	1.673	.001	.048

Table 5. $S_m(z)$ for m = 20.



(a) $S_3(z)$



(b) $S_4(z)$

Figure 1. $S_m(z)$ for m = 3, 4.



(a) $S_{17}(z)$



(b) $S_{17}(z)$. Expanded Scale

Figure 2. $S_m(z)$ for m = 17.



(a) $S_{18}(z)$



(b) $S_{18}(z)$. Expanded Scale

Figure 3. $S_m(z)$ for m = 18.

APPENDIX A

From Gradshteyn and Ryzhik,[8]

$$\int_0^{\pi} e^{i\beta x} \sin^{\nu-1}(x)\,dx = \frac{\pi\, e^{i\beta\frac{\pi}{2}}}{2^{\nu-1}\,\nu\, B\left(\frac{\nu+\beta+1}{2}, \frac{\nu-\beta+1}{2}\right)} \tag{A-1}$$

for $\mathrm{Re}(\nu) > -1$.

The beta function is related to the gamma function by[9]

$$B(x,y) = \frac{\Gamma(x)\,\Gamma(y)}{\Gamma(x+y)} = B(y,x) \tag{A-2}$$

so that

$$B\left(\frac{\nu+\beta+1}{2}, \frac{\nu-\beta+1}{2}\right) = \frac{\Gamma\left(\frac{\nu+\beta+1}{2}\right)\Gamma\left(\frac{\nu-\beta+1}{2}\right)}{\Gamma(\nu+1)} \tag{A-3}$$

and

$$\int_0^{\pi} e^{i\beta x} \sin^{\nu-1}(x)\,dx = \frac{\pi\, e^{i\beta\frac{\pi}{2}}\,\Gamma(\nu+1)}{2^{\nu-1}\,\nu\,\Gamma\left(\frac{\nu+\beta+1}{2}\right)\Gamma\left(\frac{\nu-\beta+1}{2}\right)} \tag{A-4}$$

Note,

$$\Gamma(\nu+1) = \nu\,\Gamma(\nu) \tag{A-5}$$

---

[8] Gradshteyn, I. S., and I. W. Ryzhik, Table of Integrals, Series and Products, Academic Press, New York, 1965, p. 476, Eq. 3.892.1.

[9] Ref. (8), p. 950, Eq. 8.384.1.

APPENDIX B

From Gradshteyn and Ryzhik,[10]

$$\int_0^{\pi/2} e^{i2\beta x} \sin^{2\mu}(x)\,dx = \frac{1}{2^{2\mu+1}} \left\{ \exp\left(i\pi\left(\beta-\frac{1}{2}\right)\right) B(\beta-\mu\,, 1) \right.$$

$$F(-2\mu\,, \beta-\mu\,; 1+\beta-\mu\,; -1) + \exp\left(i\pi\left(\mu+\frac{1}{2}\right)\right)$$

$$\left. B(\beta-\mu\,, 2\mu+1)\, F(0, \beta-\mu\,; 1+\beta+\mu\,; -1) \right\} \qquad \text{(B-1)}$$

for $\mathrm{Re}(\mu) > -\frac{1}{2}$.

The series expansion of a hypergeometric function is,[11]

$$F(\alpha, \beta; \gamma; z) = 1 + \frac{\alpha \cdot \beta}{\gamma \cdot 1} z + \frac{\alpha(\alpha+1)\beta(\beta+1)}{\gamma(\gamma+1)\cdot 1\cdot 2} z^2 + \frac{\alpha(\alpha+1)(\alpha+2)\beta(\beta+1)(\beta+2)}{\gamma(\gamma+1)(\gamma+2)\cdot 1\cdot 2\cdot 3} z^3 + \ldots \qquad \text{(B-2)}$$

for $\gamma \neq 0$.

Since in this case $1 + \beta + \mu \neq 0$,

$$F(0, \beta-\mu\,; 1+\beta+\mu\,; -1) = 1 \qquad \text{(B-3)}$$

Eq. (B-1) is not valid for $\mu = -\frac{1}{2}$ since,

$$B(\beta-\mu\,, 2\mu+1) = B(\beta+\frac{1}{2}, 0) = \Gamma(0) = \infty \qquad \text{(B-4)}$$

However, consider $2\mu = \epsilon - 1$ which satisfies the condition that $\mathrm{Re}(\mu) > -\frac{1}{2}$, for $\epsilon > 0$.

---

[10] Ref (8), p. 476, Eq. 3.892.3.

[11] Ref (8), p. 1039, Eq. 9.100.

So,

$$\int_0^{\pi/2} e^{i2\beta x} \sin^{\epsilon - 1}(x)\, dx = \frac{1}{2^{\epsilon}} \Big\{ \exp(i\pi(\beta - \tfrac{1}{2}))\, B(\beta - \tfrac{\epsilon}{2} + \tfrac{1}{2}, 1)$$

$$F(1-\epsilon, \beta - \tfrac{\epsilon}{2} + \tfrac{1}{2}; \beta - \tfrac{\epsilon}{2} + \tfrac{1}{2} + 1; -1)$$

$$+ \exp\left(\frac{i\pi\epsilon}{2}\right) B(\beta - \tfrac{\epsilon}{2} + \tfrac{1}{2}, \epsilon)\Big\} \qquad \text{(B-5)}$$

Since the beta function, B, and the hypergeometric function, F, are real functions the complex conjugate of the whole equation may be taken. So,

$$\int_0^{\pi/2} e^{-i2\beta x} \sin^{\epsilon - 1}(x)\, dx = \left[\int_0^{\pi/2} e^{i2\beta x} \sin^{\epsilon - 1}(x)\, dx\right]^* \qquad \text{(B-6)}$$

$$= \frac{1}{2^{\epsilon}} \Big\{ \exp(-i\pi(\beta - \tfrac{1}{2}))\, B(\beta - \tfrac{\epsilon}{2} + \tfrac{1}{2}, 1)$$

$$F(1-\epsilon, \beta - \tfrac{\epsilon}{2} + \tfrac{1}{2}; \beta - \tfrac{\epsilon}{2} + \tfrac{1}{2} + 1; -1)$$

$$+ \exp\left(\frac{-i\pi\epsilon}{2}\right) B(\beta - \tfrac{\epsilon}{2} + \tfrac{1}{2}, \epsilon)\Big\}$$

Therefore,

$$I_1 = \lim_{\epsilon \to 0} \int_0^{\pi/2} \left(e^{i2\beta_1 x} - e^{-i2\beta_2 x}\right) \sin^{\epsilon - 1}(x)\, dx$$

$$= \lim_{\epsilon \to 0} 2^{-\epsilon} \Big\{ \exp(i\pi(\beta_1 - \tfrac{1}{2}))\, B(\beta_1 - \tfrac{\epsilon}{2} + \tfrac{1}{2}, 1)$$

$$F(1-\epsilon, \beta_1 - \tfrac{\epsilon}{2} + \tfrac{1}{2}; \beta_1 - \tfrac{\epsilon}{2} + \tfrac{1}{2} + 1; -1)$$

$$- \exp(-i\pi(\beta_2 - \frac{1}{2}))\, B(\beta_2 - \frac{\epsilon}{2} + \frac{1}{2}, 1)\, F(1-\epsilon, \beta_2 - \frac{\epsilon}{2} + \frac{1}{2}; \beta_2 - \frac{\epsilon}{2} + \frac{1}{2} + 1; -1)$$

$$+ \exp\left(\frac{i\pi\epsilon}{2}\right) B(\beta_1 - \frac{\epsilon}{2} + \frac{1}{2}, \epsilon) - \exp\left(\frac{-i\pi\epsilon}{2}\right) B(\beta_2 - \frac{\epsilon}{2} + \frac{1}{2}, \epsilon)\Big\} \qquad \text{(B-7)}$$

Consider the key terms,

$$T = \lim_{\epsilon \to 0} \left\{ \exp\left(\frac{i\pi\epsilon}{2}\right) B(\beta_1 - \frac{\epsilon}{2} + \frac{1}{2}, \epsilon) - \exp\left(-\frac{i\pi\epsilon}{2}\right) B(\beta_2 - \frac{\epsilon}{2} + \frac{1}{2}, \epsilon) \right\}$$

$$= \lim_{\epsilon \to 0} \left\{ \exp\left(\frac{i\pi\epsilon}{2}\right) \frac{\Gamma(\beta_1 - \frac{\epsilon}{2} + \frac{1}{2})\Gamma(\epsilon)}{\Gamma(\beta_1 - \frac{\epsilon}{2} + \frac{1}{2} + \epsilon)} - \exp\left(-\frac{i\pi\epsilon}{2}\right) \frac{\Gamma(\beta_2 - \frac{\epsilon}{2} + \frac{1}{2})\Gamma(\epsilon)}{\Gamma(\beta_2 - \frac{\epsilon}{2} + \frac{1}{2} + \epsilon)} \right\}$$

$$= \lim_{\epsilon \to 0} \frac{\Gamma(1+\epsilon)}{\epsilon} \left\{ \cos\left(\frac{\pi\epsilon}{2}\right) \left[\frac{\Gamma(\beta_1 + \frac{1+\epsilon}{2} - \epsilon)}{\Gamma(\beta_1 + \frac{1+\epsilon}{2})} - \frac{\Gamma(\beta_2 + \frac{1+\epsilon}{2} - \epsilon)}{\Gamma(\beta_2 + \frac{1+\epsilon}{2})} \right] \right.$$

$$\left. + i \sin\left(\frac{\pi\epsilon}{2}\right) \left[\frac{\Gamma(\beta_1 + \frac{1+\epsilon}{2} - \epsilon)}{\Gamma(\beta_1 + \frac{1+\epsilon}{2})} + \frac{\Gamma(\beta_2 + \frac{1+\epsilon}{2} - \epsilon)}{\Gamma(\beta_2 + \frac{1+\epsilon}{2})} \right] \right\} \qquad \text{(B-8)}$$

By using a Taylor expansion,

$$T = \frac{\Gamma'(\beta_2 + \frac{1}{2})}{\Gamma(\beta_2 + \frac{1}{2})} - \frac{\Gamma'(\beta_1 + \frac{1}{2})}{\Gamma(\beta_1 + \frac{1}{2})} + i\pi \qquad \text{(B-9)}$$

$$= \psi(\beta_2 + \frac{1}{2}) - \psi(\beta_1 + \frac{1}{2}) + i\pi$$

where the psi-function,[12]

$$\psi(\alpha) = \frac{d}{d\alpha} \ln\Gamma(\alpha) = \frac{\Gamma'(\alpha)}{\Gamma(\alpha)} \qquad \text{(B-10)}$$

---

[12] Ref. (8), p. 943, Eq. 8.360.

It can also be shown that

$$B(\alpha, 1) = \frac{\Gamma(\alpha)\,\Gamma(1)}{\Gamma(\alpha+1)} = \frac{\Gamma(\alpha)}{\alpha\Gamma(\alpha)} = \frac{1}{\alpha} \tag{B-11}$$

and,[13]

$$F(1, \alpha; 1+\alpha; -1) = \frac{1}{2}\alpha\left[\psi(\frac{1}{2}+\frac{\alpha}{2}) - \psi(\frac{\alpha}{2})\right] \tag{B-12}$$

Let

$$G(\alpha) = \left[\psi\ (\frac{1}{2}+\frac{\alpha}{2}) - \psi(\frac{\alpha}{2})\right] \tag{B-13}$$

Finally, one obtains,

$$\begin{aligned} I_1 &= \frac{1}{2}e^{i\pi(\beta_1-\frac{1}{2})}\left[\psi\left(\frac{\beta_1}{2}+\frac{3}{4}\right) - \psi\left(\frac{\beta_1}{2}+\frac{1}{4}\right)\right] - \frac{1}{2}e^{-i\pi(\beta_2-\frac{1}{2})}\left[\psi\left(\frac{\beta_2}{2}+\frac{3}{4}\right)\right. \\ &\quad \left. - \psi\left(\frac{\beta_2}{2}+\frac{1}{4}\right)\right] + \left[\psi\,(\beta_2+\frac{1}{2}) - \psi(\beta_1+\frac{1}{2})\right] + i\pi \\ &= \frac{1}{2}\left[e^{i\pi(\beta_1-\frac{1}{2})}G(\beta_1+\frac{1}{2}) - e^{-i\pi(\beta_2-\frac{1}{2})}G(\beta_2+\frac{1}{2})\right] + \left[\psi(\beta_2+\frac{1}{2}) - \psi(\beta_1+\frac{1}{2})\right] + i\pi \end{aligned} \tag{B-14}$$

Similarly,

$$I_2 = \lim_{\epsilon\to 0}\int_0^{\pi/2}\left(e^{-i2\beta_1 x} - e^{-i2\beta_2 x}\right)\sin^{\epsilon-1}(x)\,dx$$

$$= \lim_{\epsilon\to 0} 2^{-\epsilon}\left\{\exp(-i\pi\,(\beta_1-\frac{1}{2}))\,B(\beta_1-\frac{\epsilon}{2}+\frac{1}{2}, 1)\,F(1-\epsilon, \beta_1-\frac{\epsilon}{2}+\frac{1}{2}; \beta_1-\frac{\epsilon}{2}+\frac{1}{2}+1; -1)\right.$$

[13] Ref. (2), p. 557, Eq. 15.1.23.

$$- \exp\left[-i\pi(\beta_2 - \tfrac{1}{2})\right] B(\beta_2 - \tfrac{\epsilon}{2} + \tfrac{1}{2}, 1)\, F(1-\epsilon, \beta_2 - \tfrac{\epsilon}{2} + \tfrac{1}{2}; \beta_2 - \tfrac{\epsilon}{2} + \tfrac{1}{2} + 1; -1)$$

$$+ \exp\left(-\frac{i\pi\epsilon}{2}\right) B(\beta_1 - \tfrac{\epsilon}{2} + \tfrac{1}{2}, \epsilon) - \exp\left(-\frac{i\pi\epsilon}{2}\right) B(\beta_2 - \tfrac{\epsilon}{2} + \tfrac{1}{2}, \epsilon)\Big\}$$

$$= \frac{1}{2}\left[e^{-i\pi(\beta_1 - \frac{1}{2})} G(\beta_1 + \tfrac{1}{2}) - e^{-i\pi(\beta_2 - \frac{1}{2})} G(\beta_2 + \tfrac{1}{2})\right] + \left[\psi(\beta_2 + \tfrac{1}{2}) - \psi(\beta_1 + \tfrac{1}{2})\right] \qquad \text{(B-15)}$$

APPENDIX C

```
      COMPLEX FUNCTION SM(M,X)
C
C     SM(M,Z) COMPUTES THE INTEGRAL OF THE ANGER-WEBER FUNCTION BY
C     SERIES, INTEGRATION, OR ASYSPTOTIC FORM DEPENDING ON THE VALUES
C     OF M AND Z.
C
      COMMON /SW/ SWITCH(100),ISWFLAG,DIF
      COMPLEX J
      DOUBLE PRECISION Z,PML(350),C,S1,S2,Z2,ZL,T1,T2,A1,A2,A(250),B(250
     1),SQPI
      DATA MSAVE/-1/,NUM/188/,J/0.,1./
      DATA PI,OP,PI02/3.141592653589793,.318309886183791,1.570796326795/
      DATA IFLAG/-1/,SQPI/1.772453850905516027298167D0/
      COMMON /NT/ N
      COMMON /GQ/ RM,TX,NFLAG,K
      IF (IFLAG) 10,50,50
C
C     COMPUTE AND STORE GAMMA FUNCTIONS.
C
   10 A(12)=SQPI
      DO 20 I=1,11
   20 A(12-I)=DBLE(-2./FLOAT(2*I-1))*A(13-I)
      DO 30 I=1,175
   30 A(12+I)=DBLE(.5*FLOAT(2*I-1))*A(11+I)
      B(1)=1.D0
      B(2)=B(1)
      DO 40 I=2,175
   40 B(I+1)=DBLE(FLOAT(I))*B(I)
      IFLAG=1
C
C     CALL SWITCH ROUTINE TO STORE VALUES AT WHICH TO CHANGE FROM THE
C     INTEGRATION METHOD TO THE ASYMPTOTIC FORM FOR ERR=1 PERCENT AND
C     MAXIMUM M=20 IF ROUTINE HAS NOT BEEN PREVIOUSLY CALLED BY USER.
C
      IF (ISWFLAG.NE.1) CALL ACCUR (20,.01)
   50 IF (MSAVE-M) 60,130,60
C
C     COMPUTE AND STORE COEFFICIENTS.
C
   60 C=0.
      DO 120 IL=1,NUM
      L=IL-1
      IF (MOD(L+M,2)) 70,80,70
   70 A1=A((L+M+25)/2)
      A2=A((L-M+25)/2)
      GO TO 110
   80 NAG=(L-M+2)/2
      IF (NAG) 90,90,100
   90 PML(IL)=0.
      GO TO 120
  100 A1=B(NAG)
      A2=B((L+M+2)/2)
  110 IF (DABS(C).GT.1.D300) GO TO 90
      C=DBLE (FLOAT(IL))*A1*A2
      PML(IL)=1.D0/C
  120 CONTINUE
      MSAVE=M
      NU=NUM-3
```

```
130   CONTINUE
C
C     COMPUTE FUNCTION BY SERIES APPROXIMATION.
C
      S1=0.
      S2=0.
      IF (X.LE.0.) GO TO 180
      IF (X.GE.33.) GO TO 160
      Z=DBLE(X)
      Z2=Z*Z
      DO 140 IL=1,NU,4
      N=IL+3
      ZL=Z**IL
      T1=(PML(IL)-PML(IL+2)*Z2)*ZL
      T2=(PML(IL+1)-PML(IL+3)*Z2)*ZL*Z
      S1=S1+T1
      S2=S2+T2
      IF (S1.EQ.0.) GO TO 140
      IF (DABS(T1/S1).LT.1.D-7) GO TO 150
140   CONTINUE
150   S3=S1
      S4=S2
      SM=CMPLX(S4,S3)*J**M
      RETURN
160   RM=FLOAT(M)
      TX=2.*X
      IF (X.GT.SWITCH(M+1))  GO TO 170
C
C     COMPUTE FUNCTION BY INTEGRATION.
C
      K=M
      NFLAG=-1
      CALL GQINT (0,0.,PI02,.0001,S1,DUM)
      NFLAG=1
      CALL GQINT (0,0.,PI02,.0001,S2,DUM)
      SM=OP*CMPLX(S1,S2)
      IF (MOD(M,2).NE.0) SM=J*SM
      RETURN
C
C     COMPUTE FUNCTION BY ASYMPTOTIC APPROXIMATION.
C
170   AL=.5*OP*(ALOG(X-.5*(RM-1.))+((-1.)**M)*ALOG(X+.5*(RM+1.))-(1.+((-
     11.)**M))*PSI(.5*(RM+1.)))
      SM=CMPLX(AL,.5)-CMPLX(.5/SQRT(PI*X),0.)*CEXP(CMPLX(0.,(.5*RM+.25)*
     1PI-TX))
      RETURN
180   SM=(0.,0.)
      RETURN
      END
```

```
      SUBROUTINE GQINT (IPRN,XL,XU,E,SUM,ERSM)
C
C     NUMERICAL INTEGRATION BY GAUSSIAN QUADRATURE METHOD OF ORDER 40
C
      DIMENSION R(40), U(40)
      DATA M/40/,U/-.998237709710559,-.990726238699457,-.977259949983774
     1,-.957916819213792,-.932812808278677,-.902098806968874,-.865959503
     2212260,-.824612230833312,-.778305651426519,-.727318255189927,-.671
     3956684614180,-.612553889667980,-.549467125095128,-.483075801686179
     4,-.413779204371605,-.341994090825758,-.268152185007254,-.192697580
     5701371,-.116084070675255,-.387724175060508E-1,.387724175060508E-1,
     6.116084070675255,.192697580701371,.268152185007254,.34199409082575
     78,.413779204371605,.483075801686179,.549467125095128,.612553889667
     8980,.671956684614180,.727318255189927,.778305651426519,.8246122308
     933312,.865959503212260,.902098806968874,.932812808278677,.95791681
     $9213792,.977259949983774,.990726238699457,.998237709710559/
      DATA R/.452127709853319E-2,.104982845311528E-1,.164210583819079E-1
     1,.222458491941670E-1,.279370069800234E-1,.334601952825478E-1,.3878
     221679744720E-1,.438709081856733E-1,.486958076350722E-1,.5322784698
     339368E-1,.574397690993916E-1,.613062424929289E-1,.648040134566010E
     4-1,.679120458152339E-1,.706116473912868E-1,.728865823958041E-1,.74
     57231690579683E-1,.761103619006262E-1,.770398181642480E-1,.77505947
     69784248E-1,.775059479784248E-1,.770398181642480E-1,.76110361900626
     72E-1,.747231690579683E-1,.728865823958041E-1,.706116473912868E-1,.
     8679120458152339E-1,.648040134566010E-1,.613062424929289E-1,.574397
     9690993916E-1,.532278469839368E-1,.486958076350722E-1,.438709081856
     $733E-1,.387821679744720E-1,.334601952825478E-1,.279370069800234E-1
     $,.222458491941670E-1,.164210583819079E-1,.104982845311528E-1,.4521
     $27709853319E-2/
      DATA NPRT/256/
      N=1
      CHK=0.0
   10 SUM=0.0
      XN=N
      DLX=(XU-XL)/XN
      HDLX=0.5*DLX
      AI=XL
      DO 30 I=1,N
      ANS=0.0
      AIP1=AI+DLX
      ASM=AI+AIP1
      HASM=0.5*ASM
      XX=HDLX*U(1)+HASM
      DO 20 J=1,M
      FX=FOFX(XX)
      ANS=ANS+FX*R(J)
   20 XX=HDLX*U(J+1)+HASM
      SUM=SUM+HDLX*ANS
   30 AI=AIP1
      IF (IPRN) 40,50,40
   40 PRINT 90, SUM,N
   50 ERSM=SUM-CHK
      IF (ABS(ERSM/SUM)-E) 80,80,60
```

```
60  N=2*N
    CHK=SUM
    IF (N-NPRT) 10,10,70
70  CONTINUE
80  RETURN
C
90  FORMAT (27H APPROX. VALUE OF INTEGRAL=E21.14,4H FORI4,12H PARTITIO
   1NS.)
    END
```

```
      FUNCTION FOFX (X)
C
C     COMPUTES THE INTEGRAND OF THE SM(M,Z) FUNCTION, THE INTEGRAL OF
C     THE ANGER-WEBER FUNCTION.  THE INTEGRAND IS SEPARATED INTO REAL
C     AND IMAGINARY PARTS FOR EVEN AND ODD M.
C
      COMMON /GQ/ RM,TZ,NFLAG,M
      SX=SIN(X)
      C1=COS(RM*X)/SX
      C2=SIN(RM*X)/SX
      IF (NFLAG) 10,10,40
   10 IF (MOD(M,2)) 30,20,30
   20 FOFX=C1*(1.-COS(TZ*SX))
      RETURN
   30 FOFX=C2*(1.-COS(TZ*SX))
      RETURN
   40 IF (MOD(M,2)) 60,50,60
   50 FOFX=C1*SIN(TZ*SX)
      RETURN
   60 FOFX=C2*SIN(TZ*SX)
      RETURN
      END
```

```
      SUBROUTINE ACCUR (N, ERR)
C
C     SEARCH AND FIND Z VALUES AT WHICH THE RELATIVE DIFFERENCE BETWEEN
C     THE FUNCTION SM AS GENERATED BY THE ASYMPTOTIC FORMULA AND BY THE
C     INTEGRATION METHOD IS LESS THAN OR EQUAL TO ERR.  THIS WILL FIND
C     THESE VALUES FOR ALL M FROM 0 TO N.  THE VALUES FOR ERR=.01
C     (1 PERCENT) ARE STORED FOR M=0 TO M=20 AND THIS IS THE DEFAULT
C     OPTION.
C
      DIMENSION SAM20E1(21)
      COMMON /GQ/ RM,TX,NFLAG,M
      COMMON /SW/ SWITCH(100),ISWFLAG,DIF
      DATA SAM20E1/33.,34.,35.,37.,40.,45.156,45.63,75.47,58.42,105.78,7
     15.47,136.09,105.78,181.56,120.94,211.87,136.09,257.34,166.41,302.8
     21,196.72/
      IF (N.EQ.20.AND.ERR.EQ..01) GO TO 20
      DIF=ERR
      MM=N+1
      DO 10 I=1,MM
      M=I-1
      RM=FLOAT(M)
      CALL ZERO (30.,1000.,.001,100,X,IDUM)
      SWITCH(I)=X
      IF (IDUM.EQ.-4) SWITCH (I)=33.+FLOAT(M)
   10 CONTINUE
      ISWFLAG=1
      RETURN
   20 ISWFLAG=1
      DO 30 I=1,21
   30 SWITCH(I)=SAM20E1(I)
      RETURN
      END
```

```
      SUBROUTINE ZERO (XL,XU,ACC,ITER,XO,IFLAG)
C
C     THIS SUBROUTINE CALCULATES THE ABSCISSA VALUE FOR THE ZERO OF A
C     FUNCTION FX(X) BETWEEN XL AND XU.
C
C     INPUT PARAMETERS ARE AS FOLLOWS
C           XL   - LOWER VALUE OF RANGE IN WHICH THE ZERO IS SOUGHT
C           XU   - UPPER VALUE OF RANGE IN WHICH THE ZERO IS SOUGHT
C           ACC  - ABSOLUTE ACCURACY TO WHICH ZERO IS CALCULATED
C           ITER- MAXIMUM NUMBER OF ITERATIONS DESIRED IN THE CALCULATIONS
C     ALSO THE FUNCTION FX(X) MUST BE PROVIDED
C     OUTPUT PARAMETERS ARE AS FOLLOWS
C           XO     - THE ABSCISSA VALUE AT THE ZERO OF THE FUNCTION
C           IFLAG - FLAG INDICATING THE FOLLOWING
C                 -1     FX(XL)=0,        XO IS SET TO XL
C                 -2     FX(XU)=0,        XO IS SET TO XU
C                 -3     FX(XL) AND FX(XU)=0,          XO IS SET TO XL
C                 -4     FX(XL) AND FX(XU) BOTH POSITIVE OR NEGATIVE
C                        INDICATING NO ZERO OR EVEN NUMBER OR ZEROS BETWEEN
C                        XL AND XU,    XO IS SET TO 1.0E+99
C                 -5     NUMBER OF ITERATIONS EXCEED INPUT MAXIMUM (ITER)
C                        XO IS SET TO LAST VALUE CALCULATED
C                 IF IFLAG IS ANY POSITIVE NUMBER THIS IS THE NUMBER OF
C                 ITERATIONS USED BY THE ROUTINE TO FIND XO.
      I1=0
      I2=0
      I=0
      A=XL
      B=XU
      Y1=FX(A)
      Y2=FX(B)
      IF (ABS(Y1)-ACC) 10,10,20
   10 I1=-1
   20 IF (ABS(Y2)-ACC) 30,30,40
   30 I2=-2
   40 IFLAG=I1+I2
      IF (IFLAG) 50,60,60
   50 IF (IFLAG+2) 150,160,150
   60 IF (Y1*Y2) 80,80,70
   70 IFLAG=-4
      XO=1.E99
      RETURN
   80 I=I+1
      IF (ITER-I) 140,90,90
   90 X=(A+B)*.5
      Y=FX(X)
      IF (ABS(Y)-ACC) 130,130,100
  100 IF (Y*Y1) 120,130,110
  110 A=X
      GO TO 80
  120 B=X
      GO TO 80
```

```
130   XO=X
      IFLAG=I
      RETURN
140   IFLAG=-5
      XO=X
      RETURN
150   XO=XL
      RETURN
160   XO=XU
      RETURN
      END
```

```
      FUNCTION FX (X)
C
C     CALCULATE RELATIVE DIFFERENCE BETWEEN INTEGRATION AND ASYMPTOTIC
C     FORMULAS TO DETERMINE SWITCHING POINTS.
C
      COMPLEX INT,AM
      COMMON /VAL/ INT,AM
      COMMON /SW/ SWITCH(100),ISWFLAG,DIF
      COMMON /GQ/ RM,TX,NFLAG,M
      DATA PI,OP,PIO2/3.141592653589793,.318309886183791,1.570796326795/
      TX=2.*X
      NFLAG=-1
      CALL GQINT (0,0.,PI02,.0001,S1,DUM)
      NFLAG=1
      CALL GQINT (0,0.,PI02,.0001,S2,DUM)
      INT=OP*CMPLX(S1,S2)
      IF (MOD(M,2).NE.0) INT=(0.,1.)*INT
   10 AL=.5*OP*(ALOG(X-.5*(RM-1.))+((-1.)**M)*ALOG(X+.5*(RM+1.))-(1.+((-
     11.)**M))*PSI(.5*(RM+1.)))
      AM=CMPLX(AL,.5)-CMPLX(.5/SQRT(PI*X),0.)*CEXP(CMPLX(0.,(.5*RM+.25)*
     1PI-TX))
      DI=CABS((AM-INT)/INT)
      FX=DIF-DI
      RETURN
      END
```

```
      FUNCTION PSI (Z)
C
C     COMPUTES PSI FUNCTION PSI(Z+1)=PSI(Z) + 1/Z,      Z REAL
C     AND AN ODD MULTIPLE OF 1/2.  THE INITIAL VALUE IS
C     PSI(1/2)=-1.963510026021423, STORED IN PS(1).  PS(2) CONTAINS
C     PSI(3/2), ETC.
C
      DIMENSION PS(400)
      DATA IFLAG/1/
      IF (IFLAG) 30,30,10
   10 PS(1)=-1.963510026021423
      DO 20 I=2,400
      OZ=2./FLOAT(2*I-3)
      PS(I)=PS(I-1)+OZ
   20 CONTINUE
      IFLAG=-1
   30 N=IFIX(Z+.500001)
      PSI=PS(N)
      RETURN
      END
```